

Sensitivity Analysis Using Different Equivalence Scales

Goal

The LIS equivalence scale (square root of the number of household members) is just one among many possible equivalence scales. Obviously, the choice of the scale will have an impact on the measure calculated. The difference is all the more important when you are considering specific subgroups of the population that are treated differently by the different equivalence scales. In this exercise we will see how the head count ratio (HCR) changes when using the OECD equivalence scales (modified and original) with respect to the LIS equivalence scale.

Activity

Calculate the head count ratio (HCR) for the total population, the child population, and the elderly population in Finland in 2000. Use the LIS equivalence scale, the OECD modified scale, and the OECD original scale to calculate three different measures of equivalised income.

Guidelines

- This exercise utilizes three different equivalence scales:
 - LIS scale = square root of the number of persons;
 - OECD modified scale = $1 + 0.5 * \text{number of other adult members} + 0.3 * \text{number of children below 14}$;
 - OECD original scale = $1 + 0.7 * \text{number of other adult members} + 0.5 * \text{number of children below 14}$.
- To calculate the head count ratio with the three equivalence scales for the three population groups:
 - Calculate three different equivalised incomes:
 - `%LET scale = dpi/SQRT(d4) ;`
 - `%LET scale = dpi/(1 + 0.5*(d4-1-d29) + 0.3*d29) ;`
 - `%LET scale = dpi/(1 + 0.7*(d4-1-d29) + 0.5*d29) ;`

Program

```
OPTIONS NOSOURCE NONOTES NOFMterr NODATE NOCENTER LABEL NONUMBER LS=MAX
PS=MAX ;
```

```
%MACRO avg ;
```

```
  PROC MEANS DATA=scale MEAN ;
```

```
    VAR    poor    ;
```

```
    WEIGHT &poids ;
```

```
  RUN;
```

```
%MEND avg ;
```

```
%MACRO senana ;
```

```
  DATA scale ;
```

```
    SET &fi00h (KEEP=hweight d4 d5 d29 num6574 numge75 dpi);
```

```
    IF dpi in (. 0) THEN DELETE;
```

```
    IF d5 =3 THEN DELETE;
```

```
    ey = &scale ;
```

```
    wt = hweight * d4 ;
```

```
    cwt = hweight * d29;
```

```
    ewt = hweight * SUM(num6574,numge75);
```

```
  RUN;
```

```
  PROC UNIVARIATE DATA=scale NOPRINT;
```

```
    VAR ey;
```

```
    WEIGHT wt ;
```

```
    OUTPUT OUT=temp MEDIAN=medey;
```

```
  RUN ;
```

```
  DATA _NULL_;
```

```
    SET temp;
```

```
    CALL SYMPUT("m",medey);
```

```
  RUN;
```

```
  DATA scale (KEEP=poor wt cwt ewt) ;
```

```
    SET scale ;
```

```
    poor    = 0 ;
```

```
    eymed   = &m ;
```

```
    povlin  = &m * 0.5 ;
```

```
    IF ey < povlin THEN poor = 1 ;
```

```
  RUN ;
```

```
  TITLE2 "***** TOTAL POPULUATION *****" ;
```

```
  %LET poids = wt ;
```

```
  %avg
```

```
  TITLE2 "***** CHILDREN *****" ;
```

```
  %LET poids = cwt ;
```

```
  %avg
```

```
  TITLE2 "***** ELDERLY *****" ;
```

```
  %LET poids = ewt ;
```

```
  %avg
```

```
%MEND senana;
```

```
TITLE1 "LIS Equivalence Scale" ;
```

```
%LET scale = dpi/SQRT(d4);
```

```
%senana
```

```
TITLE1 "OECD Modified Scale" ;
```

```
%LET scale = dpi/(1 + 0.5*(d4-1-d29) + 0.3*d29);
```

```
%senana
TITLE1 "OECD Original Scale" ;
%LET scale = dpi/(1 + 0.7*(d4-1-d29) + 0.5*d29);
%senana
```

Results

HCR	LIS equivalence scale	OECD modified scale	OECD original scale
Total population	5.432	4.324	3.872
Children (< 14)	3.257	3.104	5.591
Elderly	8.469	4.736	1.179

Comments

- As expected, the HCR changes when using different equivalence scales. This is especially true for child and elderly poverty rates.
- The poverty rate changes substantially between different equivalence scales. Equivalised income ey is calculated as:

$$ey = y / d^\varepsilon$$

where y is the unequivalised disposable income (**dpi** in LIS terms), d is the size of the family (**d4** in LIS terms) and ε is the equivalence elasticity. ε varies between 0 (indicating full economies of scale and no need to adjust household income with size) and 1 (no economies of scale, which indicates the need to use per capita income). Obviously, the larger the elasticity, the smaller the economies of scale assumed by the equivalence scale. HCR increases with ε for large households, whereas it decreases for small households. This is because with a higher ε (i.e., smaller economies of scale), large households will need more income to get the same standards of living, whereas smaller households need less income. The LIS equivalence scale uses $\varepsilon = 0.5$; the original OECD scale corresponds to a value of $\varepsilon \approx 0.7$; and the modified OECD scale stands somewhere in the middle. As a result, when using the OECD scales, poverty will generally be higher in households with children (which tend to be larger than the average household), and much lower in households with elderly (which tend to be much smaller).